
du sens

comment définir la signification des programmes écrits dans un langage ?

la plupart du temps, on se contente d’une description informelle, en langue
naturelle (norme ISO, standard, ouvrage de référence, etc.)

s’avère peu satisfaisant, car souvent imprécis, voire ambigu

sémantique informelle

The Java programming language
guarantees that the operands of
operators appear to be evaluated in
a specific evaluation order, namely,
from left to right.

It is recommended that code not
rely crucially on this specification.

Jean-Christophe Filliâtre Langages de programmation et compilation sémantique 3



sémantique formelle

la sémantique formelle caractérise mathématiquement les calculs décrits
par un programme

utile pour la réalisation d’outils (interprètes, compilateurs, etc.)

indispensable pour raisonner sur les programmes

amène une autre question

mais qu’est-ce qu’un programme ?

en tant qu’objet syntaxique (suite de caractères),
il est trop difficile à manipuler

on préfère utiliser la syntaxe abstraite

syntaxe abstraite

source
↓

analyse lexicale
↓

suite de lexèmes (tokens)
↓

analyse syntaxique
↓

syntaxe abstraite
↓

analyse sémantique
↓

syntaxe abstraite

↓
production de code

↓
langage assembleur

↓
assembleur (as)

↓
langage machine

↓
éditeur de liens (ld)

↓
code exécutable

syntaxe abstraite

les textes

2*(x+1)

et

(2 * ((x) + 1))

et

2 * (* je multiplie par deux *) ( x + 1 )

représentent tous le même arbre de syntaxe abstraite

×

2 +

x 1

notation

on définit une syntaxe abstraite par une grammaire, de la manière suivante

e ::= c constante
| x variable
| e + e addition
| e × e multiplication
| ...

se lit ≪ une expression e est

• soit une constante,

• soit une variable,

• soit l’addition de deux expressions,

• etc. ≫

notation

la notation e1 + e2 de la syntaxe abstraite emprunte le symbole de la
syntaxe concrète

mais on aurait pu tout aussi bien choisir add(e1, e2), +(e1, e2), etc.

représentation de la syntaxe abstraite

en OCaml, on réalise la syntaxe abstraite par des types construits

type binop = Add | Mul | ...

type expression =

| Cte of int

| Var of string

| Bin of binop * expression * expression

| ...

l’expression 2 * (x + 1) est représentée par

Bin (Mul, Cte 2, Bin (Add, Var "x", Cte 1))

parenthèses

il n’y a pas de constructeur dans la syntaxe abstraite pour les parenthèses

dans la syntaxe concrète 2 * (x + 1),
les parenthèses servent à reconnâıtre cet arbre

×

2 +

x 1
plutôt que

+

×

2 x

1

(on expliquera comment dans un autre cours)

sucre syntaxique

on appelle sucre syntaxique une construction de la syntaxe concrète qui
n’existe pas dans la syntaxe abstraite

elle est donc traduite à l’aide d’autres constructions de la syntaxe abstraite
(généralement à l’analyse syntaxique)

exemples :

• en OCaml, l’expression [e1; e2; ...; en] est du sucre pour

e1 :: e2 :: ... :: en :: []

• en C, l’expression a[i] est du sucre pour *(a+i)

sémantique

c’est sur la syntaxe abstraite que l’on va définir la sémantique

il existe de nombreuses approches

• sémantique axiomatique

• sémantique dénotationnelle

• sémantique par traduction

• sémantique opérationnelle

sémantique axiomatique

encore appelée logique de Floyd–Hoare
(Robert Floyd, Assigning meanings to programs, 1967
Tony Hoare, An axiomatic basis for computer programming, 1969)

caractérise les programmes par l’intermédiaire des propriétés satisfaites par
les variables ; on introduit le triplet

{P} i {Q}

signifiant ≪ si la formule P est vraie avant l’exécution de l’instruction i ,
alors la formule Q sera vraie après ≫

exemple :
{x ≥ 0} x := x + 1 {x > 0}

exemple de règle :

{P[x ← E ]} x := E {P(x)}
Jean-Christophe Filliâtre Langages de programmation et compilation sémantique 14



sémantique dénotationnelle

la sémantique dénotationnelle associe à chaque expression e sa
dénotation [[e]], qui est un objet mathématique représentant le calcul
désigné par e

exemple : expressions arithmétiques avec une seule variable x

e ::= x | n | e + e | e * e | . . .

la dénotation peut être une fonction qui associe à la valeur de x la valeur
de l’expression

[[x]]
def
= x 7→ x

[[n]]
def
= x 7→ n

[[e1 + e2]]
def
= x 7→ [[e1]](x) + [[e2]](x)

[[e1 * e2]]
def
= x 7→ [[e1]](x)× [[e2]](x)

sémantique par traduction

(encore appelée sémantique dénotationnelle à la Strachey)

on peut définir la sémantique d’un langage en le traduisant vers un
langage dont la sémantique est déjà connue

exemple

un langage ésotérique dont la syntaxe est composée de huit caractères et
dont la sémantique peut être définie par traduction vers le langage C

commande traduction en C

(prélude) char array[30000] = {0};

char *ptr = array;

> ++ptr;

< --ptr;

+ ++*ptr;

- --*ptr;

. putchar(*ptr);

, *ptr = getchar();

[ while (*ptr) {

] }

sémantique opérationnelle

la sémantique opérationnelle décrit l’enchâınement des calculs
élémentaires qui mènent de l’expression à son résultat (sa valeur)

deux formes de sémantique opérationnelle

• ≪ sémantique naturelle ≫ ou ≪ à grands pas ≫ (big-steps semantics)

e ↠ v

• ≪ sémantique à réductions ≫ ou ≪ à petits pas ≫ (small-steps
semantics)

e → e1 → e2 → · · · → v

mini-ML

illustrons la sémantique opérationnelle sur le langage mini-ML

e ::= x variable
| c constante (1, 2, . . ., true, . . .)
| op primitive (+, ×, fst, . . .)
| fun x → e fonction
| e e application
| (e, e) paire
| let x = e in e liaison locale

exemples

let compose = fun f → fun g → fun x → f (g x) in
let plus = fun x → fun y → + (x , y) in
compose (plus 2) (plus 4) 36

let distr pair = fun f → fun p → (f (fst p), f (snd p)) in
let p = distr pair (fun x → x) (40, 2) in
+ (fst p, snd p)

autres constructions : conditionnelle

la conditionnelle peut être définie comme

if e1 then e2 else e3
def
= opif (e1, ((fun → e2), (fun → e3)))

où opif est une primitive

les branches sont gelées à l’aide de fonctions

autres constructions : récursivité

de même, la récursivité peut être définie comme

rec f x = e
def
= opfix (fun f → fun x → e)

où opfix est un opérateur de point fixe, satisfaisant

opfix f = f (opfix f )

exemple d’utilisation :

opfix (fun fact → fun n→ if n = 0 then 1
else × (n, fact (−(n, 1))))

sémantique opérationnelle à grands pas de mini-ML

on cherche à définir une relation entre une expression e et une valeur v

e ↠ v

les valeurs sont ainsi définies

v ::= c constante
| op primitive non appliquée
| fun x → e fonction
| (v , v) paire

pour définir e ↠ v , on a besoin des notions de règles d’inférence
et de substitution

Jean-Christophe Filliâtre Langages de programmation et compilation sémantique opérationnelle 23



règles d’inférence

une relation peut être définie comme la plus petite relation satisfaisant
un ensemble d’axiomes de la forme

P

et un ensemble d’implications de la forme

P1 P2 . . . Pn

P

règles d’inférence

exemple : on peut définir la relation Pair(n) par les deux règles

Pair(0)
et

Pair(n)

Pair(n + 2)

qui doivent se lire comme

d’une part Pair(0)
et d’autre part ∀n. Pair(n)⇒ Pair(n + 2)

la plus petite relation satisfaisant ces deux propriétés cöıncide avec la
propriété ≪ n est un entier naturel pair ≫ :

• les entiers pairs sont clairement dedans, par récurrence

• s’il y avait un entier impair, on pourrait enlever le plus petit

arbre de dérivation

une dérivation est un arbre dont les nœuds correspondent aux règles et les
feuilles aux axiomes ; exemple

Pair(0)
Pair(2)

Pair(4)

l’ensemble des dérivations possibles caractérise exactement la plus petite
relation satisfaisant les règles d’inférence

Jean-Christophe Filliâtre Langages de programmation et compilation sémantique opérationnelle 26



variables libres

Définition (variables libres)

L’ensemble des variables libres d’une expression e, noté fv(e), est défini
par récurrence sur e de la manière suivante :

fv(x) = {x}
fv(c) = ∅

fv(op) = ∅
fv(fun x → e) = fv(e) \ {x}

fv(e1 e2) = fv(e1) ∪ fv(e2)
fv((e1, e2)) = fv(e1) ∪ fv(e2)

fv(let x = e1 in e2) = fv(e1) ∪ (fv(e2) \ {x})

Une expression sans variable libre est dite close.

exemples

fv (let x = +(20, 1) in (fun y → +(y , y)) x) = ∅

fv (let x = z in (fun y → (x y) t)) = {z , t}

substitution

Définition (substitution)

Si e est une expression, x une variable et v une valeur, on note e[x ← v ]
la substitution de toute occurrence libre de x dans e par v , définie par

x [x ← v ] = v
y [x ← v ] = y si y ̸= x
c[x ← v ] = c

op[x ← v ] = op
(fun x → e)[x ← v ] = fun x → e
(fun y → e)[x ← v ] = fun y → e[x ← v ] si y ̸= x

(e1 e2)[x ← v ] = (e1[x ← v ] e2[x ← v ])
(e1, e2)[x ← v ] = (e1[x ← v ], e2[x ← v ])

(let x = e1 in e2)[x ← v ] = let x = e1[x ← v ] in e2
(let y = e1 in e2)[x ← v ] = let y = e1[x ← v ] in e2[x ← v ]

si y ̸= x

exemples

((fun x → +(x , x)) x) [x ← 21] = (fun x → +(x , x)) 21

(+(x, let x = 17 in x)) [x ← 3] = +(3, let x = 17 in x)

(fun y → y y) [y ← 17] = fun y → y y

sémantique naturelle de mini-ML

c ↠ c op ↠ op (fun x → e) ↠ (fun x → e)

e1 ↠ v1 e2 ↠ v2
(e1, e2) ↠ (v1, v2)

e1 ↠ v1 e2[x ← v1] ↠ v

let x = e1 in e2 ↠ v

e1 ↠ (fun x → e) e2 ↠ v2 e[x ← v2] ↠ v

e1 e2 ↠ v

note : on a fait le choix d’une stratégie d’appel par valeur
i.e. l’argument est complètement évalué avant l’appel (cf cours 7)

primitives

il faut ajouter des règles pour les primitives ; par exemple

e1 ↠ + e2 ↠ (n1, n2) n = n1 + n2
e1 e2 ↠ n

e1 ↠ opif e2 ↠ (true, ((fun → e3), (fun → e4))) e3 ↠ v

e1 e2 ↠ v

e1 ↠ opfix e2 ↠ (fun f → e) e[f ← opfix (fun f → e)] ↠ v

e1 e2 ↠ v

e1 ↠ fst e2 ↠ (v1, v2)

e1 e2 ↠ v1

exemple de dérivation

+ ↠ +
20↠20 1↠1

(20, 1) ↠ (20, 1)

+(20, 1) ↠ 21

fun...↠fun... 21↠21

...

+(21, 21) ↠ 42

(fun y → +(y , y)) 21 ↠ 42

let x = +(20, 1) in (fun y → +(y , y)) x ↠ 42

exercice

donner la dérivation de
(opfix F ) 2

avec F défini comme

fun fact → fun n→ if n = 0 then 1 else × (n, fact (−(n, 1)))

expressions sans valeur

il existe des expressions e pour lesquelles il n’y a pas de valeur v
telle que e ↠ v

exemple : e = 1 2

exemple : e = (fun x → x x) (fun x → x x)

récurrence sur la dérivation

pour établir une propriété d’une relation définie par un ensemble de règles
d’inférence, on peut raisonner par récurrence sur la dérivation

cela signifie par récurrence structurelle i.e. on peut appliquer l’hypothèse
de récurrence à toute sous-dérivation
(de manière équivalente, on peut dire que l’on raisonne par récurrence sur
la hauteur de la dérivation)

en pratique, on raisonne par récurrence sur la dérivation et par cas sur la
dernière règle utilisée

propriétés de la sémantique naturelle de mini-ML

Proposition (l’évaluation produit des valeurs closes)

Si e ↠ v alors v est une valeur.
De plus, si e est close, alors v l’est également.

preuve par récurrence sur la dérivation e ↠ v
cas d’une application

(D1)
...

e1 ↠ (fun x → e)

(D2)
...

e2 ↠ v2

(D3)
...

e[x ← v2] ↠ v

e1 e2 ↠ v

par HR v est une valeur
si e est close alors e1 et e2 aussi, et par HR fun x → e et v2
sont closes, donc e[x ← v2] est close, et par HR v aussi

(exercice : traiter les autres cas)

propriétés de la sémantique naturelle de mini-ML

Proposition (déterminisme de l’évaluation)

Si e ↠ v et e ↠ v ′ alors v = v ′.

par récurrence sur les dérivations de e ↠ v et de e ↠ v ′

cas d’une paire e = (e1, e2)

(D1)
...

e1 ↠ v1

(D2)
...

e2 ↠ v2

(e1, e2) ↠ (v1, v2)

(D ′
1)
...

e1 ↠ v ′1

(D ′
2)
...

e2 ↠ v ′2
(e1, e2) ↠ (v ′1, v

′
2)

par HR on a v1 = v ′1 et v2 = v ′2 donc v = (v1, v2) = (v ′1, v
′
2) = v ′

(exercice : traiter les autres cas)

déterminisme

remarque : la relation d’évaluation n’est pas nécessairement déterministe

exemple : on ajoute une primitive random et la règle

e1 ↠ random e2 ↠ n1 0 ≤ n < n1
e1 e2 ↠ n

on a alors random 2 ↠ 0 aussi bien que random 2 ↠ 1

interprète

on peut programmer un interprète en suivant les règles de la sémantique
naturelle

on se donne un type pour la syntaxe abstraite des expressions

type expression = ...

et on définit une fonction

val eval: expression -> expression

correspondant à la relation ↠ (puisqu’il s’agit d’une fonction)

interprète pour mini-ML

type expression =

| Var of string

| Const of int

| Op of string

| Fun of string * expression

| App of expression * expression

| Pair of expression * expression

| Let of string * expression * expression

interprète pour mini-ML

il faut coder l’opération de substitution e[x ← v ]

val subst: expression -> string -> expression -> expression

on suppose v close (donc pas de problème de capture de variable)

let rec subst e x v = match e with

| Var y ->

if y = x then v else e

| Fun (y, e1) ->

if y = x then e else Fun (y, subst e1 x v)

| Let (y, e1, e2) ->

Let (y, subst e1 x v, if y = x then e2 else subst e2 x v)

| App (e1, e2) ->

App (subst e1 x v, subst e2 x v)

| Pair (e1, e2) ->

Pair (subst e1 x v, subst e2 x v)

| Const _ | Op _ ->

e

interprète pour mini-ML

la sémantique naturelle est réalisée par la fonction

val eval: expression -> expression

let rec eval = function

| Const _ | Op _ | Fun _ as v ->

v

| Pair (e1, e2) ->

Pair (eval e1, eval e2)

| Let(x, e1, e2) ->

eval (subst e2 x (eval e1))

...

interprète pour mini-ML

...

| App (e1, e2) ->

begin match eval e1 with

| Fun (x, e) ->

eval (subst e x (eval e2))

| Op "+" ->

let (Pair (Const n1, Const n2)) = eval e2 in

Const (n1 + n2)

| Op "fst" ->

let (Pair(v1, v2)) = eval e2 in v1

| Op "snd" ->

let (Pair(v1, v2)) = eval e2 in v2

end

exemple d’évaluation

# eval

(Let

("x",

App (Op "+", Pair (Const 1, Const 20)),

App (Fun ("y", App (Op "+", Pair (Var "y", Var "y"))),

Var "x")));;

- : expression = Const 42

interprète pour mini-ML

le filtrage est volontairement non-exhaustif

# eval (Var "x");;

# eval (App (Const 1, Const 2));;

Exception: Match_failure ("", 87, 6).

(on pourrait préférer un type option, une exception explicite, etc.)

interprète pour mini-ML

l’évaluation peut ne pas terminer

par exemple sur
(fun x → x x) (fun x → x x)

# let b = Fun ("x", App (Var "x", Var "x")) in

eval (App (b, b));;

Interrupted.

exercice

ajouter les opérateurs opif et opfix à cet interprète

petit intermède...

peut-on éviter l’opération de substitution ?

idée : on interprète l’expression e à l’aide d’un environnement donnant la
valeur courante de chaque variable (un dictionnaire)

val eval: environment -> expression -> value

difficulté : le résultat de

let x = 1 in fun y → +(x , y)

est une fonction qui doit ≪ mémoriser ≫ que x = 1

réponse : il faut utiliser une fermeture

petit intermède...

on utilise le module Map pour les environnements

module Smap = Map.Make(String)

on définit un nouveau type pour les valeurs

type value =

| Vconst of int

| Vop of string

| Vpair of value * value

| Vfun of string * environment * expression

and environment = value Smap.t

petit intermède...

val eval: environment -> expression -> value

let rec eval env = function

| Const n ->

Vconst n

| Op op ->

Vop op

| Pair (e1, e2) ->

Vpair (eval env e1, eval env e2)

| Var x ->

Smap.find x env

| Let (x, e1, e2) ->

eval (Smap.add x (eval env e1) env) e2

| ...

petit intermède...

| Fun (x, e) ->

Vfun (x, env, e)

| App (e1, e2) ->

begin match eval env e1 with

| Vfun (x, clos, e) ->

eval (Smap.add x (eval env e2) clos) e

| Vop "+" ->

let Vpair (Vconst n1, Vconst n2) = eval env e2 in

Vconst (n1 + n2)

| Vop "fst" ->

let Vpair (v1, _) = eval env e2 in v1

| Vop "snd" ->

let Vpair (_, v2) = eval env e2 in v2

end

note : c’est ce que l’on fait quand on compile ML (cf cours 8)

exercice

ajouter l’opérateur opif à cet interprète

note : ajouter l’opérateur opfix est plus complexe
(on en reparlera dans le cours 8)

défauts de la sémantique naturelle

la sémantique naturelle ne permet pas de distinguer les expressions dont le
calcul ≪ plante ≫, comme

1 2

des expressions dont l’évaluation ne termine pas, comme

(fun x → x x) (fun x → x x)

sémantique opérationnelle à petits pas

la sémantique opérationnelle à petits pas y remédie en introduisant une
notion d’étape élémentaire de calcul e1 → e2, que l’on va itérer

on peut alors distinguer trois situations

1. l’itération aboutit à une valeur

e → e1 → e2 → · · · → v

2. l’itération bloque sur en irréductible qui n’est pas une valeur

e → e1 → e2 → · · · → en

3. l’itération ne termine pas

e → e1 → e2 → · · ·

sémantique opérationnelle à petits pas pour mini-ML

on commence par définir une relation
ϵ→ correspondant à une réduction

≪ en tête ≫, c’est-à-dire au sommet de l’expression

deux règles :

(fun x → e) v
ϵ→ e[x ← v ]

let x = v in e
ϵ→ e[x ← v ]

note : là encore, on a fait le choix d’une stratégie d’appel par valeur

sémantique opérationnelle à petits pas pour mini-ML

on se donne également des règles pour les primitives

+ (n1, n2)
ϵ→ n avec n = n1 + n2

fst (v1, v2)
ϵ→ v1

snd (v1, v2)
ϵ→ v2

opfix (fun f → e)
ϵ→ e[f ← opfix (fun f → e)]

opif (true, ((fun → e1), (fun → e2)))
ϵ→ e1

opif (false, ((fun → e1), (fun → e2)))
ϵ→ e2

Jean-Christophe Filliâtre Langages de programmation et compilation sémantique opérationnelle 57



sémantique opérationnelle à petits pas pour mini-ML

pour réduire en profondeur, on introduit la règle d’inférence

e1
ϵ→ e2

E (e1)→ E (e2)

où E est un contexte, défini par la grammaire suivante :

E ::= □
| E e
| v E
| let x = E in e
| (E , e)
| (v ,E )

contextes

un contexte est donc un ≪ terme à trou ≫, où □ représente le trou

exemple :

E
def
= let x = +(2,□) in let y = +(x , x) in y

E (e) dénote le contexte E dans lequel □ a été remplacé par e

e

exemple :

E (+(10, 9)) = let x = +(2,+(10, 9)) in let y = +(x , x) in y

réduction dans un contexte

la règle

e1
ϵ→ e2

E (e1)→ E (e2)

permet donc d’évaluer une sous-expression
e1 e2

exemple : on a la réduction

+(1, 2)
ϵ→ 3

let x = +(1, 2) in + (x , x)→ let x = 3 in + (x , x)

grâce au contexte E
def
= let x = □ in + (x , x)

ordre d’évaluation

tels qu’on les a choisis, les contextes impliquent ici une évaluation en appel
par valeur et ≪ de gauche à droite ≫

ainsi, (+(1, 2),□) n’est pas un contexte d’évaluation

on aurait pu également choisir une évaluation de droite à gauche

définitions

on note
⋆→ la clôture réflexive et transitive de →

(i.e. e1
⋆→ e2 ssi e1 se réduit en e2 en zéro, une ou plusieurs étapes)

on appelle forme normale toute expression e telle qu’il n’existe pas
d’expression e ′ telle que e → e ′

les valeurs sont des formes normales ; les formes normales qui ne sont pas
des valeurs sont les expressions erronées (comme 1 2)

programmer la sémantique à réductions

on va écrire les fonctions suivantes :

val head_reduction: expression -> expression

correspond à
ϵ→

val decompose: expression -> context * expression

décompose une expression sous la forme E (e)
avec e réductible en tête

val reduce1: expression -> expression option

correspond à →

val reduce: expression -> expression

correspond à
⋆→

programmer la sémantique à réductions

on commence par caractériser les valeurs

let rec is_a_value = function

| Const _ | Op _ | Fun _ ->

true

| Var _ | App _ | Let _ ->

false

| Pair (e1, e2) ->

is_a_value e1 && is_a_value e2

programmer la sémantique à réductions

on écrit ensuite la réduction en tête

let head_reduction = function

| App (Fun (x, e1), e2) when is_a_value e2 ->

subst e1 x e2

| Let (x, e1, e2) when is_a_value e1 ->

subst e2 x e1

| App (Op "+", Pair (Const n1, Const n2)) ->

Const (n1 + n2)

| App (Op "fst", Pair (e1, e2))

when is_a_value e1 && is_a_value e2 ->

e1

| App (Op "snd", Pair (e1, e2))

when is_a_value e1 && is_a_value e2 ->

e2

| _ ->

raise NoReduction

programmer la sémantique à réductions

un contexte E peut être directement représenté par la fonction e 7→ E (e)

type context = expression -> expression

let hole = fun e -> e

let app_left ctx e2 = fun e -> App (ctx e, e2)

let app_right v1 ctx = fun e -> App (v1, ctx e)

let pair_left ctx e2 = fun e -> Pair (ctx e, e2)

let pair_right v1 ctx = fun e -> Pair (v1, ctx e)

let let_left x ctx e2 = fun e -> Let (x, ctx e, e2)

programmer la sémantique à réductions

let rec decompose e = match e with

(* on ne peut décomposer *)

| Var _ | Const _ | Op _ | Fun _ ->

raise NoReduction

(* cas d’une réduction en tête *)

| App (Fun (x, e1), e2) when is_a_value e2 ->

(hole, e)

| Let (x, e1, e2) when is_a_value e1 ->

(hole, e)

| App (Op "+", Pair (Const n1, Const n2)) ->

(hole, e)

| App (Op ("fst" | "snd"), Pair (e1, e2))

when is_a_value e1 && is_a_value e2 ->

(hole, e)

...

programmer la sémantique à réductions

...

(* cas d’une réduction en profondeur *)

| App (e1, e2) ->

if is_a_value e1 then

let (ctx, rd) = decompose e2 in

(app_right e1 ctx, rd)

else

let (ctx, rd) = decompose e1 in

(app_left ctx e2, rd)

| Let (x, e1, e2) ->

let (ctx, rd) = decompose e1 in

(let_left x ctx e2, rd)

| Pair (e1, e2) ->

...

programmer la sémantique à réductions

let reduce1 e =

try

let ctx, e’ = decompose e in

Some (ctx (head_reduction e’))

with NoReduction ->

None

enfin

let rec reduce e =

match reduce1 e with None -> e | Some e’ -> reduce e’

efficacité

un tel interprète n’est pas très efficace

il passe son temps à recalculer le contexte puis à l’≪ oublier ≫

on peut faire mieux, par exemple en utilisant un zipper [Huet, 1997]

équivalence des deux sémantiques opérationnelles

nous allons montrer que les deux sémantiques opérationnelles sont
équivalentes pour les expressions dont l’évaluation termine sur une valeur,
i.e.

e ↠ v si et seulement si e
⋆→ v

équivalence des deux sémantiques opérationnelles

Lemme (passage au contexte des réductions)

Supposons e → e ′. Alors pour toute expression e2 et toute valeur v

1. e e2 → e ′ e2

2. v e → v e ′

3. let x = e in e2 → let x = e ′ in e2

preuve : de e → e ′ on sait qu’il existe un contexte E tel que

e = E (r) e ′ = E (r ′) r
ϵ→ r ′

considérons le contexte E1
def
= E e2 ; alors

r
ϵ→ r ′

E1(r)→ E1(r ′)
i .e.

r
ϵ→ r ′

e e2 → e ′ e2

(de même pour les cas 2 et 3)

équivalence des deux sémantiques opérationnelles

Proposition (“grands pas” implique “petits pas”)

Si e ↠ v, alors e
⋆→ v.

preuve : par récurrence sur la dérivation de e ↠ v
supposons que la dernière règle soit

e1 ↠ (fun x → e3) e2 ↠ v2 e3[x ← v2] ↠ v

e1 e2 ↠ v

équivalence des deux sémantiques opérationnelles

par HR on a
e1 → · · · → v1 = (fun x → e3)
e2 → · · · → v2
e3[x ← v2]→ · · · → v

par passage au contexte (lemme précédent) on a également

e1 e2 → · · · → v1 e2
v1 e2 → · · · → v1 v2
e3[x ← v2]→ · · · → v

en insérant la réduction

(fun x → e3) v2
ϵ→ e3[x ← v2]

on obtient la réduction complète

e1 e2 → · · · → v

□

équivalence des deux sémantiques opérationnelles

pour le sens inverse (“petits pas” implique “grands pas”) on a besoin de
deux lemmes

Lemme (les valeurs sont déjà évaluées)

v ↠ v pour toute valeur v .

preuve : immédiat

équivalence des deux sémantiques opérationnelles

Lemme (réduction et évaluation)

Si e → e ′ et e ′ ↠ v, alors e ↠ v.

preuve : on commence par les réductions de tête i.e. e
ϵ→ e ′

supposons par exemple e = (fun x → e1) v2 et e ′ = e1[x ← v2]
on construit la dérivation

(fun x → e1) ↠ (fun x → e1) v2 ↠ v2 e1[x ← v2] ↠ v

(fun x → e1) v2 ↠ v

en utilisant le lemme précédent (v2 ↠ v2) et l’hypothèse e ′ ↠ v

(traiter les autres cas)

équivalence des deux sémantiques opérationnelles

montrons maintenant que si e
ϵ→ e ′ et E (e ′) ↠ v alors E (e) ↠ v

par récurrence structurelle sur E ; on vient de faire le cas E = □

considérons le cas E = E ′ e2
on a E (e ′) ↠ v c’est-à-dire E ′(e ′) e2 ↠ v , qui a par exemple la forme

E ′(e ′) ↠ (fun x → e3) e2 ↠ v2 e3[x ← v2] ↠ v

E ′(e ′) e2 ↠ v

par HR on a E ′(e) ↠ (fun x → e3), et donc

E ′(e) ↠ (fun x → e3) e2 ↠ v2 e3[x ← v2] ↠ v

E ′(e) e2 ↠ v

c’est-à-dire E (e) ↠ v (traiter les autres cas) □

équivalence des deux sémantiques opérationnelles

Proposition (“petits pas” implique “grands pas”)

Si e
⋆→ v, alors e ↠ v.

preuve : supposons e → e1 → · · · → en → v
on a v ↠ v et donc par le lemme précédent en ↠ v
de même en−1 ↠ v
et ainsi de suite jusqu’à e ↠ v

quid des langages impératifs ?

on peut définir une sémantique opérationnelle, à grands pas ou à petits
pas, pour un langage avec des traits impératifs

on associe typiquement un état S à l’expression évaluée / réduite

S , e ↠ S ′, v ou encore S , e → S ′, e ′

exemple de règle :

S , e ↠ S ′, v

S , x := e ↠ S ′ ⊕ {x 7→ v}, void

l’état S peut être décomposé en plusieurs éléments, pour modéliser
une pile (des variables locales), un tas, et plus encore...

la suite

• TD 2
• écriture d’un interprète

pour mini-Python

• prochain cours
• analyse lexicale

> ./mini-python tests/good/pascal.py

*

**

***

****

*****

******

*******

*000000*

**00000**

***0000***

****000****

*****00*****

******0******

**************

*000000*000000*

**00000**00000**

***0000***0000***

****000****000****

*****00*****00*****

******0******0******

*********************

